

UNITED STATES DEPARTMENT OF THE INTERIOR
GEOLOGICAL SURVEY

A program for calculating and plotting uranium-lead concordia-discordia intercepts using the Hewlett-Packard Series 80 personal computers and the HP-7470A graphic plotter

by

John J. Brice III¹

Open-File Report 84-478

This report is preliminary and has not been reviewed for conformity with U.S. Geological Survey editorial standards and stratigraphic nomenclature. Any use of trade names is for descriptive purposes only and does not imply endorsement by the U.S. Geological Survey.

¹. Menlo Park, California

Contents

	<u>Page</u>
Introduction -----	1
Equipment required -----	3
The program -----	3
Running the program -----	7
Program list -----	8
Appendix I -----	15

Illustrations

Figure 1. Concordia-discordia test graph plotted from data in

Appendix I -----	2
------------------	---

A program for calculating and plotting uranium-lead concordia-discordia
intercepts using the Hewlett-Packard Series 80 personal computers and
the HP-7470A graphic plotter

By John J. Brice III

Introduction

This program is in HP-series 80 Basic and is a convenient method for calculating and tabulating concordia-discordia results and graphically presenting U-Pb isotope data on a concordia curve diagram. The concordia curve diagram can be plotted as a high quality plot on either standard paper suitable for reproduction or on another media such as transparency film.

The concordia curve is the locus of $^{206}\text{Pb}^*/^{238}\text{U}$ versus $^{207}\text{Pb}^*/^{235}\text{U}$ points, for which the $^{206}\text{Pb}^*/^{238}\text{U}$ and $^{207}\text{Pb}^*/^{235}\text{U}$ apparent ages are equal. A useful characteristic of the concordia curve diagram is that rocks or minerals that have either undergone varying degrees of no more than one episodic disturbance in their Pb*/U ratio or have lost varying amounts of Pb by a process of continuous diffusion, will define a straight line, called the discordia line. This line normally intersect the concordia curve at two points (fig. 1). The upper intercept with the concordia curve may be related to the primary age of the zircons present. The lower intercept may be related to episodic lead loss and to the time elapsed since original crystallization of the minerals (the age of the rock).

The user of this program enters a set of $^{206}\text{Pb}^*/^{238}\text{U}$ and $^{207}\text{Pb}^*/^{235}\text{U}$ data points for each sample, a maximum of 100 samples can be entered. The program then prints (see appendix 1) the upper intercept's $^{206}\text{Pb}^*/^{238}\text{U}$ and $^{207}\text{Pb}^*/^{235}\text{U}$ quotidiants and the corresponding age in millions of years. Then the lower intercept and age is printed, and the correlation coefficient for the discordia line (the best fit-line) and the slope of the line and the




Figure 1.--Concordia-discordia test graph plotted from data in Appendix I.

Y-intercept are printed. Then the user is asked if a graphical plot of the U-Pb isotope data on a concordia curve diagram is required. If the answer is yes the computer will draw the graph on the 7470A Plotter. The computer will draw the graph with a automatic scale (determined by the longest data point or the upper intercept whichever is greater) or the user may enter the minimum and maximum for the scales and the computer will then draw the graph to that scale, plus 25 percent.

Equipment required

The Hewlett-Packard Series 80 Personal Computer uses HP Series 80 Basic. This program was written on the HP 85 computer and to enhance the capacity of the computer an Advanced Programing ROM, Plotter/Printer ROM, and Matrix ROM was incorporated into the unit. The Hewlett-Packard 7470A Graphic Plotter, a vector plotter, was used to produce the graph on standard paper and also on other media such as transparency film.

The program

The program has been dimensioned to accept up to 100 sets of data points. This number maybe expanded or decreased by changing the X() and Y() on line 90. When the computer requests the $^{206}\text{Pb}^*/^{238}\text{U}$ fraction number, the user enters the number and then presses the End Line key. The computer will request more data sets until the 100 points maximum is achieved or the user signals an end by entering a negative $^{206}\text{Pb}^*/^{238}\text{U}$ fraction number.

The program was written for the HP 85 computer and the print statements with Tabs will need to be changed for a full line printer. The Tab lines are 250, 260, 270, 380. Lines 40, 60, 80, and 1490 will also have to be changed. There will be a spacing problem between words on lines 1930 and 3080. The spacing will also have to be changed on line 2160 for the display screen.

After all data points are entered the program will display "Program is working" as the computer calculates the coefficients, determines the largest $^{207}\text{Pb}^*/^{235}\text{U}$ data point to scale the graph, and the intercept points with the concordia curve. Then the computer prints the upper and lower intercept's age in millions of years and the 7*/5 and 6*/8 location of both intercepts. The age scale can be changed by changing the equations on lines 950 and 1090 and by changing scale labeling on lines 1130, 1240, 2000, 2540, and 2720. After the information on the intercept points is printed, information on the best-fit concordia line is printed. The correlation coefficient number is then printed. This is the proportion of the variance of the data points explained by the best-fit concordia line. Then the slope and the Y-intercept of the best-fit line are printed.

The program can be ended at this point, or the user may proceed to the plotting portion of the program. The plotting portion consists of two steps: autograph and self-graph. If the autograph step is requested, five prescaled graphs will be plotted. These can be used to get synoptic view of all of the data points. If the user then would like to study one portion of the graph, the graph can be rescaled to a scale that the user chooses by using the self-graph step. The user is requested to enter the scaling information, and then the computer will redraw the graph to that scale. If some or all of the data points fall outside of the new scaled boundaries they will not be plotted.

Before the user proceeds to the self-graph step, the user may want to remove the printout from the printer. When the self-graph plot section of the program is activated the printer will print out the information needed for graphing.

If the user does not want to use the autograph step, the user types no when the program asks if you want to autograph. The program will then print:

```
NOTE: THERE IS A BUILT-IN 25%  
INCREASE IN THE 7*/5 AND 6*/8  
MAX. SCALE POINTS.
```

```
IF YOU ENTER A WRONG NUMBER  
PRESS PAUSE AND THEN CONT.  
THEN REENTER THE NUMBERS FROM  
N(1,1).
```

```
N(1,1)=7*/5 MAX. POINT  
N(1,2)=7*/5 MIN. POINT  
N(1,3)=7*/5 TIC SPACING  
N(1,4)=7*/5 MAJOR TIC SPACING  
N(1,5)=FXD OF 7*/5  
N(1,6)=TIME TIC SPACING IN m.v.  
N(1,7)=FXD OF 6*/8  
N(1,8)=6*/8 MAX. POINT  
N(1,9)=6*/8 MIN. POINT  
N(1,10)=6*/8 TIC SPACING
```

This information explains what the user is required to enter. If the user does not want to use the 25 percent increase in the 7*/5 and 6*/8 maximum scale points on the graph. Lines 2310 and 2370 should be changed. If 25 percent is eliminated the $N(I,1) \times 1.25$ in both lines and the $N(I,8) \times 1.25$ in line 2310 should be changed to $N(I,1)$ and $N(I,8)$. Line 1930 can then be eliminated. If the user enters the wrong data, press the Pause and then the Continue keys.

This will erase the plotting information, and the display will show:

N(1,1)? After the N(1,10) data is entered the program plots the graph. The self-graph can be used more than once, but the printout (see above) is only printed once.

In both the autograph and the self-graph steps you can title your graph with a maximum of 50 characters. If no title is needed press the Return Key and the program will proceed to the scaling of the graph.

This program provides a quick convenient way to plot uranium-lead isotope data points on a concordia curve diagram and to determine the age of a rock. The plotting portion of the program can graphicly display the data for quick evaluation. This program has not been designed to handle errors and error-correlations.

Running the Program

Execute the RUN command.

<u>Display</u>	<u>Response</u>
ENTER FRACTION # N (TO TERMINATE, ENTER 6*/8<0)	
6*/8?	Enter your $^{206}\text{Pb}^*/^{238}\text{U}$ fraction number. After your last data points have been entered and this request is displayed again enter any negative number. The program then proceeds to the next step.
7*/5?	Enter $^{207}\text{Pb}^*/^{235}\text{U}$ fraction number.
PROGRAM IS WORKING	
DO YOU WANT A GRAPH?	Enter YES or NO
DO YOU WANT AN AUTO PLOTTED GRAPH?	Enter YES or NO
ENTER THE FOLLOWING DATA	
N(1,N)?	See the printout for the needed response. N=1 thru 10
ENTER TITLE OF PLOT (50 CHAR. LONG)	If you want to title your plot, if no, press the return key.
DO YOU WANT THE FRACTION POINTS NUMBERED?	Enter YES or NO. If YES, the points will be numbered the same as they were entered.
DO YOU WANT THE DISCORDIA LINE?	Enter YES or NO
DO YOU WANT TO PLOT ANOTHER GRAPH?	Enter YES or NO. If YES, and if you have been using the autograph step, remove your printer data from the printer. If you have been using the self-graph step, the display will show "Enter the following data". This will also be displayed if you have used the autograph step and typed in yes.
DO YOU WANT TO PLOT A NEW SET OF FRACTION POINTS?	Enter YES or NO. If NO the program ends, if YES the program will start all over again. Note: remove all printed matter from printer before pressing the return key.

Program list

```

10 PRINT @ PRINT
20 OPTION BASE 1
30 CLEAR @ DISP
40 PRINT "*****"
   PRINT "*****"
50 PRINT
60 PRINT "URANIUM-LEAD CONCORDI
   A-DISCORDIAINTERCEPTS"
70 PRINT
80 PRINT "*****"
   PRINT "*****"
90 DIM X(100),Y(100),T$(58),N(5
   ,10),C(100),Y$(3),N$(2)
100 Y$="YES"
110 N$="NO"
120 T$=" ABCDEFGHIJKLMNOPQRSTUVWXYZ.0123456789!%&'(<);:,<>?+-
   */^=@"
130 ! S=S
140 ! SX=S1
150 ! SY=S2
160 ! SX2=S3
170 ! SXY=S4
180 ! SY2=S5
190 ! XI=X
200 ! YI=Y
210 ! NPTS=N
220 ! VARNCE=V
230 N=0
240 PRINT @ PRINT @ PRINT
250 PRINT "FRAC #";TAB(10);"206P
   b*";TAB(22);"207Pb*"
260 PRINT TAB(10);"-----";TAB(2
   2);"-----"
270 PRINT TAB(11);"238U";TAB(23)
   ;"235U"
280 PRINT
290 FOR K=1 TO 100
300 DISP "ENTER FRACTION #";K
310 DISP "(TO TERMINATE,ENTER 6*
   /8<0)"
320 DISP
330 DISP "6*/8 ";@ INPUT Y(K)
340 IF Y(K)<0 THEN N=K-1 @ GOTO
   410
350 DISP
360 DISP "7*/5 ";@ INPUT X(K)
370 DISP
380 PRINT TAB(2);K;TAB(10);Y(K);
   TAB(22);X(K)
390 PRINT
400 NEXT K
410 ! ACCUMULATE SUMS
420 CLEAR @ DISP
430 DISP @ DISP @ DISP "PROGRAM
   IS WORKING"
440 S=0
450 S1=0
460 S2=0
470 S3=0

```

```

480 S4=0
490 S5=0
500 FOR I=1 TO N
510 X=X(I)
520 Y=Y(I)
530 S=S+1
540 S1=S1+X
550 S2=S2+Y
560 S3=S3+X*X
570 S4=S4+X*Y
580 S5=S5+Y*Y
590 NEXT I
600 ! CALCULATE COEFFICIENTS
610 ! DELTA=D
620 D=S*S3-S1*S1
630 A=(S3*S2-S1*S4)/D
640 B=(S4*S-S1*S2)/D
650 R=(S*S4-S1*S2)/SQR(D*(S*S5-S
2*S2))
660 ! BUBBLE SORT TO DETERMINE
THE LARGEST X TO SCALE THE
GRAPH.
670 FOR K=1 TO N
680 NEXT K
690 FOR K=1 TO N
700 C(K)=X(K)
710 NEXT K
720 H=N
730 FOR J=1 TO H-1
740 FOR K=1 TO H-J
750 IF C(K)>C(K+1) THEN 790
760 T=C(K)
770 C(K)=C(K+1)
780 C(K+1)=T
790 NEXT K
800 NEXT J
810 FOR K=1 TO N
820 NEXT K
830 O=C(1)
840 PRINT @ PRINT @ PRINT
850 ! E=B IN Y=MX+B
860 X=0
870 Y=0
880 E=A-B*X
890 ! LOWER INTERCEPT
900 Y=B*X+E
910 X1=EXP(.00098485*LOG(Y+1)/.0
00155125)-1
920 D=ABS(X1-X)
930 X=X1
940 IF D>.000001 THEN 900
950 J=X
960 M=Y
970 T(1)=LOG(M+1)/.000155125
980 X=J*1.2
990 IF X<0 THEN X=.00000099
1000 Y=0
1010 ! UPPER INTERCEPT
1020 Y=EXP(.000155125*LOG(X+1)/.
00098485)-1

```

```

1030 X1=(Y-E)/B
1040 D=ABS(X1-X)
1050 X=X1
1060 Y=B*X+E
1070 IF D>.000001 THEN 1020
1080 G=X
1090 H=Y
1100 CLEAR @ DISP
1110 T(2)=LOG(H+1)/.000155125
1120 PRINT "UPPER INTERCEPT IS:"
1130 PRINT
1140 PRINT USING 1150 ; T(2)
1150 IMAGE "T=",2X,SDC3D.D," m.y
      "
1160 PRINT
1170 PRINT USING 1180 ; G
1180 IMAGE "7*/5= ",3D.5D
1190 PRINT
1200 PRINT USING 1210 ; H
1210 IMAGE "6*/8= ",3D.5D
1220 PRINT @ PRINT
1230 PRINT "LOWER INTERCEPT IS:"
1240 PRINT
1250 PRINT USING 1260 ; T(1)
1260 IMAGE "T=",2X,SDC3D.D," m.y
      "
1270 PRINT
1280 PRINT USING 1290 ; J
1290 IMAGE "7*/5= ",3D.5D
1300 PRINT
1310 PRINT USING 1320 ; M
1320 IMAGE "6*/8= ",3D.5D
1330 PRINT
1340 PRINT
1350 PRINT
1360 PRINT "CONCORDIA LINE:"
1370 PRINT
1380 PRINT
1390 PRINT USING 1400 ; R
1400 IMAGE "CORRELATION COEFFICI
      ENT= ",2.5D
1410 PRINT
1420 PRINT USING 1430 ; B
1430 IMAGE "SLOPE= ",2D.5D
1440 PRINT
1450 PRINT USING 1460 ; A
1460 IMAGE "Y-INTERCEPT= ",2D.5D
1470 PRINT
1480 PRINT
1490 PRINT "*****"
      "*****"
1500 PRINT
1510 PRINT
1520 PRINT
1530 PRINT
1540 PRINT
1550 CLEAR @ DISP "DO YOU WANT A
      GRAGH";@ INPUT A$
1560 IF A$=N$ THEN 3070
1570 ! GRAPH PLOTTING

```

```

1580 S=0
1590 DISP @ DISP "DO YOU WANT AN
      AUTO GRAPH";@ INPUT A$
1600 IF A$=N$ THEN 1880
1610 DATA .1,0,.005,4,3,10,3,.01
      6,0,.001
1620 DATA .3,0,.01,5,2,20,3,.043
      ,0,.005
1630 DATA 1,0,.05,4,2,50,2,.12,0
      ,.01
1640 DATA 6,0,.2,5,1,200,2,.36,0
      ,.03
1650 DATA 100,0,5,2,0,500,2,1,0,
      .1
1660 MAT READ N
1670 IF G>0 THEN 1730
1680 IF O>6*1.2 THEN 1860
1690 IF O>1 AND O<=6*1.2 THEN 18
      40
1700 IF O>.3 AND O<=1.2 THEN 182
      0
1710 IF O>.1 AND O<=.3*1.2 THEN
      1800
1720 IF O<=.1 THEN 1780
1730 IF G>6*1.2 THEN 1860
1740 IF G>1 AND G<=6*1.2 THEN 18
      40
1750 IF G>.3 AND G<=1.2 THEN 182
      0
1760 IF G>.1 AND G<=.3*1.2 THEN
      1800
1770 IF G<=.1 THEN 1780
1780 I=1
1790 GOTO 2120
1800 I=2
1810 GOTO 2120
1820 I=3
1830 GOTO 2120
1840 I=4
1850 GOTO 2120
1860 I=5
1870 GOTO 2120
1880 I=1
1890 REDIM N(1,10)
1900 CLEAR @ DISP @ DISP "ENTER
      THE FOLLOWING DATA"
1910 DISP
1920 IF S=1 THEN 2090
1930 PRINT "NOTE: THERE IS A BUIL
      T-IN 25% INCREASE IN THE
      7*/5 AND 6*/8 MAX. SCALE
      POINTS."
1940 PRINT @ PRINT "IF YOU ENTER
      A WRONG NUMBER PRESS P
      AUSE AND THEN CONT."
1950 PRINT "THEN REENTER THE NUM
      BERS FROM N(1,1)."
1960 PRINT
1970 PRINT "N(1,1)=7*/5 MAX. POI
      NT"

```

```

1980 PRINT "N(1,2)=7*/5 MIN. POI
      NT"
1990 PRINT "N(1,3)=7*/5 TIC SPAC
      ING"
2000 PRINT "N(1,4)=7*/5 MAJOR TI
      C SPACING"
2010 PRINT "N(1,5)=FXD OF 7*/5"
2020 PRINT "N(1,6)=TIME TIC SPAC
      ING IN m.y."
2030 PRINT "N(1,7)=FXD OF 6*/8"
2040 PRINT "N(1,8)=6*/8 MAX. POI
      NT"
2050 PRINT "N(1,9)=6*/8 MIN. POI
      NT"
2060 PRINT "N(1,10)=6*/8 TIC SPA
      CING"
2070 PRINT @ PRINT @ PRINT @ PRI
      NT
2080 S=1
2090 MAT INPUT N
2100 MAT PRINT N;
2110 PRINT @ PRINT @ PRINT @ PRI
      NT
2120 PLOTTER IS 705
2130 G=0
2140 O=0
2150 PEN 1
2160 CLEAR @ DISP @ DISP "ENTER
      TITLE OF PLOT (50 CHAR.
      LONG).";@ INPUT T$
2170 PEN 1
2180 MOVE 20,85
2190 LABEL T$
2200 PENUP
2210 MOVE 55,2
2220 LABEL "207Pb*/235U"
2230 PENUP
2240 MOVE 2,50
2250 DEG @ LDIR 90
2260 LABEL "206Pb*/238U"
2270 PENUP
2280 DEG @ LDIR 0
2290 LOCATE 14,125,15,95
2300 FRAME
2310 SCALE N(I,2),N(I,1)*1.25,N(
      I,9),N(I,8)*1.25
2320 DEG @ LDIR 90
2330 FXD N(I,5),N(I,7)
2340 LAXES N(I,3),N(I,10),N(I,2)
      ,N(I,9),N(I,4),2,5
2350 PENUP
2360 MOVE N(I,2),N(I,9)
2370 FOR X=0 TO N(I,1)*1.25 STEP
      N(I,1)/100
2380 F=.1575112961
2390 Y=(X+1)^F-1
2400 DRAW X,Y
2410 NEXT X
2420 X=0
2430 PENUP

```

```

2440 IF S=1 THEN 2630
2450 MOVE 0,0
2460 T=N(I,6)
2470 FOR L=1 TO 100
2480 Y=EXP(T*.000155125)-1
2490 X=EXP(T*.00098485)-1
2500 T=LOG(X+1)/.00098485
2510 MOVE X,Y
2520 IF X<N(I,2) THEN 2590
2530 IF X>N(I,1)*1.25 THEN 2600
2540 IF Y>N(I,8)*1.25*.9 THEN 26
00
2550 IF Y<N(I,9) THEN 2590
2560 LONG 2 @ LABEL USING "2A,DC
3D,5A" ; "--";T," m.y."
2570 PENUP
2580 T=T+N(I,6)
2590 NEXT L
2600 PENUP
2610 X=0
2620 GOTO 2790
2630 MOVE 0,0
2640 X=N(I,2)
2650 T=LOG(X+1)/.00098485
2660 FOR L=1 TO 100
2670 Y=EXP(T*.000155125)-1
2680 X=EXP(T*.00098485)-1
2690 MOVE X,Y
2700 IF L=1 THEN 2750
2710 IF Y<N(I,9) THEN 2760
2720 IF X>N(I,1)*1.25 THEN 2780
2730 IF Y>N(I,8)*1.25*.9 THEN 27
80
2740 LONG 2 @ LABEL USING "2A,DC
3D,5A" ; "--";T," m.y."
2750 PENUP
2760 T=T+N(I,6)
2770 NEXT L
2780 X=0
2790 FOR L=1 TO N
2800 MOVE X(L),Y(L)
2810 IF Y(L)<N(I,9) THEN 2870
2820 IF Y(L)>N(I,8)*1.25 THEN 28
70
2830 IF X(L)<N(I,2) THEN 2870
2840 DEG @ LDIR 0
2850 LONG 5 @ CSIZE 3,1,0
2860 LABEL "*"
2870 NEXT L
2880 DISP @ DISP "DO YOU WANT TH
E FRACTION POINTS NUMBERED"
;@ INPUT C$
2890 IF C$=N$ THEN 2980
2900 FOR L=1 TO N
2910 IF Y(L)<N(I,9) THEN 2970
2920 IF Y(L)>N(I,8)*1.25*.95 THE
N 2970
2930 IF X(L)<N(I,2) THEN 2970
2940 MOVE X(L),Y(L)
2950 LONG 2 @ CSIZE 3,1,0

```

```

2960 LABEL "--";L
2970 NEXT L
2980 PENUP
2990 CLEAR @ DISP @ DISP "DO YOU
      WANT THE DISCORDIA LINE";@
      INPUT D$
3000 IF D$=N$ THEN 3050
3010 MOVE 0,A
3020 Y=A+B*(N(I,1)*1.25)
3030 DRAW N(I,1)*1.25,Y
3040 PENUP
3050 DISP @ DISP "DO YOU WANT TO
      PLOT ANOTHER      GRAPH";@
      INPUT S$
3060 IF S$=Y$ THEN 1880
3070 CLEAR @ DISP @ DISP
3080 DISP "DO YOU WANT TO PLOT A
      NEW SET OF FRACTION POINTS
      ";@ INPUT C$
3090 IF C$=Y$ THEN 10
3100 CLEAR @ DISP
3110 END

```


Appendix I

This is an example of the printout of the data using the HP-85's built-in printer, the data was used to graph figure 1.

URANIUM-LEAD CONCORDIA-DISCORDIA INTERCEPTS

FRAC #	206Pb*	207Pb*
	----- 238U	----- 235U
1	.15256	1.7953
2	.20309	2.4599
3	.12595	1.443
4	.12092	1.3767
5	.10334	1.1413

UPPER INTERCEPT IS:

T= +1,408.9 m.y.

7*/5= 3.00520

6*/8= .24429

LOWER INTERCEPT IS:

T= +225.4 m.y.

7*/5= .24849

6*/8= .03558

CONCORDIA LINE:

CORRELATION COEFFICIENT= 0.99939

SLOPE= .07571

Y-INTERCEPT= .01676
